

The test described in this document was used in the work reported in “*Evaluation of a Statistical Procedure for Particle Size Comparisons of Pharmaceutical Aerosols (Final PQRI Report)*” by D. Christopher, W. Adams, A. Amann, C. Bertha, P. Byron, W. Doub, C. Dunbar, W. Hauck, S. Lyapustina, J. P. Mitchell, B. Morgan, S. Nichols, Z. Pan, G. J. P. Singh, T. Tougas, Y. Tsong, R. Wolff, and B. Wyka. Please refer to that original report for more information.

## Population Bioequivalence

### Definition:

Population Test (T) and population Reference (R) are two independent populations. Population Test is defined to be bioequivalent to population Reference if and only if the following condition holds true.

$$\frac{(\mu_T - \mu_R)^2 + (\sigma_T^2 - \sigma_R^2)}{\sigma_R^2} \leq \Theta_p, \text{ when } \sigma_R > \sigma_{T0} \quad (1)$$

$$\frac{(\mu_T - \mu_R)^2 + (\sigma_T^2 - \sigma_R^2)}{\sigma_{T0}^2} \leq \Theta_p, \text{ when } \sigma_R \leq \sigma_{T0} \quad (2)$$

Where,

- $\mu_T$  and  $\mu_R$ ,  $\sigma_T^2$  and  $\sigma_R^2$  are population means and variances in the log scale, respectively, for Test and Reference;
- $\sigma_{T0}^2$  is a constant representing a fixed lower limit to be used for the scaling variance (i.e., the denominator in the left side of the inequality)
- $\Theta_p$  is a constant representing the critical value used in defining the bioequivalence of Test and Reference

The left side of the above inequalities has a simple and attractive meaning: it is a good measure of the distance between the Test and Reference distributions, scaled by the Reference variability, down to a fixed lower limit.

Per *Statistical Information from the June 1999 Draft Guidance and Statistical Information for In Vitro Bioequivalence Data, August 18, 1999*,\*  $\sigma_{T0}^2$  is recommended to be at least 0.01. Calculation of the critical value,  $\Theta_p$ , is based on an allowable difference between Test and Reference mean  $\mu_T - \mu_R = \log(1.11)$ , an allowable difference between

---

\* FDA, CDER. 2003. Statistical Information from the June 1999 Draft Guidance and Statistical Information for *In vitro* Bioequivalence Data Posted on August 18, 1999. Available at: <http://www.fda.gov/cder/guidance/5383stats.pdf>. Accessed May 24, 2007

Test and Reference variances  $\sigma_T^2 - \sigma_R^2 = 0.01$ , and a lower limit of the scaling variance  $\sigma_{T0}^2 = 0.01$ . The calculated result is 2.0891.  $\Theta_p$  is also called population bioequivalence limit, and it is the most important constant in defining population bioequivalence.

### **Statistical Testing of the Inequalities:**

To facilitate statistical testing of the inequalities, terms in (1) and (2), can be rearranged to give (3) and (4) as below,

$$(\mu_T - \mu_R)^2 + \sigma_T^2 - \sigma_R^2 - \Theta_p \sigma_R^2 \leq 0, \text{ when } \sigma_R > \sigma_{T0} \quad (3)$$

$$(\mu_T - \mu_R)^2 + \sigma_T^2 - \sigma_R^2 - \Theta_p \sigma_{T0}^2 \leq 0, \text{ when } \sigma_R \leq \sigma_{T0} \quad (4)$$

In order to test whether inequality (3) or (4) is true, we can calculate the 2-sided 95% confidence interval of the point estimate of the left side of the inequalities and compare its upper confidence bound to 0. If the 95% upper confidence bound is smaller than 0, then we can conclude, with 95% confidence, that the inequalities hold true, and conclude that Test population is bioequivalent to Reference population; otherwise, we cannot conclude that Test population is bioequivalent to Reference population.

## Calculation from Sample Data:

Calculation from sample data is rather complicated. Part of the calculation is shown here using SAS version 8.02.

1. Take the logarithm of the observed values to obtain the assessment results in log scale.
2. From log-scale assessment results, calculate summary statistics including number of non-missing values, sample means, and sample variances from Test and Reference samples. Assume that from 30 assessment results of Test and Reference, we have the following summary statistics as presented in the table below, where mean\_T and mean\_R are sample means, MSB\_T and MSB\_R sample variances, and n\_T and n\_R the number of non-missing values, respectively, for Test and Reference samples.

mean_T	mean_R	MSB_T	MSB_R	n_T	n_R
4.60949	4.60689	0.000481181	0.000924588	30	30

3. From the above summary statistics, the 95% confidence interval can be calculated using the following SAS program code.

```

data ci;

/* ASSIGN VALUES FOR THE SUMMARY STATISTICS */
mean_T=4.60949;
mean_R=4.60689;
MSB_T=0.000481181;
MSB_R=0.000924588;
n_T=30;
n_R=30;

/* ASSIGN VALUES FOR THE CONSTANTS USED IN DEFINING BIOEQUIVALENCE, WHERE theta_p is  $\Theta_p$ , alpha = 0.05 */

/* FOR CALCULATING 95% CONFIDENCE INTERVAL. sig IS THE SQUAE ROOT OF  $\sigma_{T0}^2$  AND EQUAL TO 0.1 */
theta_p = 2.0891;
alpha=0.05;
sig=0.1;

/* BELOW ARE SAS STATEMENTS FOR CALCULATING THE 95% CONFIDENCE INTERVAL */
f_T=n_T-1;
f_R=n_R-1;
del = mean_T - mean_R;
a1 = tinv(1-alpha, f_R);
t = tinv(1-alpha, f_T);
M = (MSB_T/n_T + MSB_R/n_R);
C = (MSB_T/n_T)/M;
a2 = 0.0313 - (13.624/f_T) - (10.182/f_R) - (29.155/(f_T*f_R)) + (25.121/(f_T*f_R)) - (5.623/(f_R*f_R));
a3 = -0.0306 + (10.236/f_T) + (13.495/f_R) - (9.822/(f_T*f_R)) - (12.207/(f_T*f_R)) + (32.396/(f_R*f_R));
a4 = 0.0243 - (8.113/f_T) - (4.703/f_R) - (12.564/(f_T*f_R)) + (14.810/(f_T*f_R)) + (3.528/(f_R*f_R));
a5 = ((a1 + a2 + a3)/t) - 1 - a4;
V = (a1 + (a2*C) + (a3*C*C))/(1 + (a4*C) + (a5*C*C));
UCL = del + (V*sqrt(M));
LCL = del - (V*sqrt(M));
E0 = del*del;
H0 = max(LCL*LCL, UCL*UCL);
E1 = (n_T - 1) * MSB_T / n_T;
H1=f_T*E1/cinv(alpha, f_T);

```

```

E2rs=-((n_R + 1) / n_R + theta_p)*MSB_R;
H2rs=f_R*E2rs/cinv(1-alpha, f_R);
E2cs=-((n_R + 1) / n_R) * MSB_R;
H2cs=f_R*E2cs/cinv(1-alpha, f_R);
U0=(H0-E0)**2;
U1=(H1-E1)**2;
U2rs=(H2rs-E2rs)**2;
U2cs=(H2cs-E2cs)**2;
upper95CL1=(E0+E1+E2rs) + sqrt(U0+U1+U2rs);
upper95CL2=(E0+E1+E2cs-theta_p*sig*sig) + sqrt(U0+U1+U2cs);

```

4. From the above SAS program code, we can obtain two 95% confidence interval upper bounds, one is for reference-scaling, and the other is for constant-scaling. In this example, we have reference-scaling upper bound **upper95CL1** = -0.001430088 and constant-scaling upper bound **upper95CL2** = -0.020909. Since we have MSB\_R = 0.000924588 (Reference sample variance) smaller than  $\sigma_{T_0}^2 = 0.01$ , we need to compare constant-scaling upper bound **upper95CL2** = -0.020909 versus 0. In our case, since **upper95CL2** < 0, we conclude that Test is bioequivalent to Reference.